MIXED

MIXED is available in the Advanced Statistics option.

```
MIXED dependent varname [BY factor list] [WITH covariate list]

[/CRITERIA = [CIN({95** })] [HCONVERGE({0**   } {ABSOLUTE**})
                  {value}                {value} {RELATIVE  }

             [LCONVERGE({0**   } {ABSOLUTE**})] [MXITER({100**})]
                        {value} {RELATIVE  }           {n    }

             [MXSTEP({5**})] [PCONVERGE({1E-6**},{ABSOLUTE**})] [SCORING({1**})]
                     {n  }              {value } {RELATIVE  }            {n  }

             [SINGULAR({1E-12**})] ]
                       {value  }

[/EMMEANS = TABLES ({OVERALL          })]
                    {factor           }
                    {factor*factor ...}
            [WITH (covariate=value [covariate = value ...])
            [COMPARE [({factor})] [REFCAT({value})] [ADJ({LSD**     })] ]
                                          {FIRST}        {BONFERRONI}
                                          {LAST }        {SIDAK     }

[/FIXED = [effect [effect ...]] [| [NOINT] [SSTYPE({1  })] ] ]
                                                   {3**}

[/METHOD = {ML    }]
           {REML**}

[/MISSING = {EXCLUDE**}]
            {INCLUDE  }

[/PRINT = [CORB] [COVB] [CPS] [DESCRIPTIVES] [G] [HISTORY(1**)] [LMATRIX] [R]
                                                          (n  )
          [SOLUTION] [TESTCOV]]

[/RANDOM = effect [effect ...]
           [| [SUBJECT(varname[*varname[*...]])] [COVTYPE({VC**      })]]]
                                                         {covstruct+}

[/REGWGT = varname]

[/REPEATED = varname[*varname[*...]] | SUBJECT(varname[*varname[*...]])
             [COVTYPE({DIAG**    })]]
                      {covstruct†}

[/SAVE = [tempvar [(name)] [tempvar [(name)]] ...]

[/TEST[(valuelist)] =
    ['label']  effect valuelist ... [| effect valuelist ...] [divisor=value]]
    [; effect valuelist ... [| effect valuelist ...] [divisor=value]]

[/TEST[(valuelist)] = ['label'] ALL list [| list] [divisor=value]]
                      [; ALL list [| list] [divisor=value]]
```

** Default if the subcommand is omitted.

† covstruct can take the following values: AD1, AR1, ARH1, ARMA11, CS, CSH, CSR, DIAG, FA1, FAH1, HF, ID, TP, TPH, UN, UNR, VC.

This command reads the active dataset and causes execution of any pending commands. For more information, see the topic Command Order on p. 42.

Example

```
MIXED Y.
```


Overview

The MIXED procedure fits a variety of mixed linear models. The mixed linear model expands the general linear model used in the GLM procedure in that the data are permitted to exhibit correlation and non-constant variability. The mixed linear model, therefore, provides the flexibility of modeling not only the means of the data but also their variances and covariances.

The MIXED procedure is also a flexible tool for fitting other models that can be formulated as mixed linear models. Such models include multilevel models, hierarchical linear models, and random coefficient models.

Important Changes to MIXED Compared to Previous Versions

Independence of random effects. Prior to version 11.5, random effects were assumed to be independent. If you are using MIXED syntax jobs from a version prior to 11.5, be aware that the interpretation of the covariance structure may have changed. For more information, see the topic Interpretation of Random Effect Covariance Structures on p. 1261.

Default covariance structures. Prior to version 11.5, the default covariance structure for random effects was ID, and the default covariance structure for repeated effects was VC.

Interpretation of VC covariance structure. Prior to version 11.5, the variance components (VC) structure was a diagonal matrix with heterogenous variances. Now, when the variance components structure is specified on a RANDOM subcommand, a scaled identity (ID) structure is assigned to each of the effects specified on the subcommand. If the variance components structure is specified on the REPEATED subcommand, it will be replaced by the diagonal (DIAG) structure. Note that the diagonal structure has the same interpretation as the variance components structure in versions prior to 11.5.

Basic Features

Covariance structures. Various structures are available. Use multiple RANDOM subcommands to model a different covariance structure for each random effect.

Standard errors. Appropriate standard errors will be automatically calculated for all hypothesis tests on the fixed effects, and specified estimable linear combinations of fixed and random effects.

Subject blocking. Complete independence can be assumed across subject blocks.

Choice of estimation method. Two estimation methods for the covariance parameters are available.

Tuning the algorithm. You can control the values of algorithm-tuning parameters with the CRITERIA subcommand.

Optional output. You can request additional output through the PRINT subcommand. The SAVE subcommand allows you to save various casewise statistics back to the active dataset.

Basic Specification

- The basic specification is a variable list identifying the dependent variable, the factors (if any) and the covariates (if any).

- By default, MIXED adopts the model that consists of the intercept term as the only fixed effect and the residual term as the only random effect.

Subcommand Order

- The variable list must be specified first.
- Subcommands can be specified in any order.

Syntax Rules

- For many analyses, the MIXED variable list, the FIXED subcommand, and the RANDOM subcommand are the only specifications needed.
- A dependent variable must be specified.
- Empty subcommands are silently ignored.
- Multiple RANDOM subcommands are allowed. However, if an effect with the same subject specification appears in multiple RANDOM subcommands, only the last specification will be used.
- Multiple TEST subcommands are allowed.
- All subcommands, except the RANDOM and the TEST subcommands, should be specified only once. If a subcommand is repeated, only the last specification will be used.
- The following words are reserved as keywords in the MIXED procedure: BY, WITH, and WITHIN.

Examples

The following are examples of models that can be specified using MIXED:

Model 1: Fixed-Effects ANOVA Model

Suppose that *TREAT* is the treatment factor and *BLOCK* is the blocking factor.

```
MIXED Y BY TREAT BLOCK
  /FIXED = TREAT BLOCK.
```

Model 2: Randomized Complete Blocks Design

Suppose that *TREAT* is the treatment factor and *BLOCK* is the blocking factor.

```
MIXED Y BY TREAT BLOCK
  /FIXED = TREAT
  /RANDOM = BLOCK.
```

Model 3: Split-Plot Design

An experiment consists of two factors, *A* and *B*. The experiment unit with respect to *A* is *C*. The experiment unit with respect to *B* is the individual subject, a subdivision of the factor *C*. Thus, *C* is the whole-plot unit, and the individual subject is the split-plot unit.

```
MIXED Y BY A B C
```

```
  /FIXED = A B A*B
  /RANDOM = C(A).
```

Model 4: Purely Random-Effects Model

Suppose that *A*, *B*, and *C* are random factors.

```
MIXED Y BY A B C
  /FIXED = | NOINT
  /RANDOM  = INTERCEPT A B C A*B A*C B*C | COVTYPE(CS).
```

The `MIXED` procedure allows effects specified on the same `RANDOM` subcommand to be correlated. Thus, in the model above, the parameters of a compound symmetry covariance matrix are computed across all levels of the random effects. In order to specify independent random effects, you need to specify separate `RANDOM` subcommands. For example:

```
MIXED Y BY A B C
  /FIXED = | NOINT
  /RANDOM  = INTERCEPT | COVTYPE(ID)
  /RANDOM  = A | COVTYPE(CS)
  /RANDOM  = B | COVTYPE(CS)
  /RANDOM  = C | COVTYPE(CS)
  /RANDOM  = A*B | COVTYPE(CS)
  /RANDOM  = A*C | COVTYPE(CS)
  /RANDOM  = B*C | COVTYPE(CS).
```

Here, the parameters of compound symmetry matrices are computed separately for each random effect.

Model 5: Random Coefficient Model

```
MIXED
  weight  WITH time
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(5) SCORING(1)
  SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE)
  LCONVERGE(0, ABSOLUTE) PCONVERGE(0.000001, ABSOLUTE)
  /FIXED = time  | SSTYPE(3)
  /METHOD = REML
  /PRINT = G SOLUTION TESTCOV
  /RANDOM INTERCEPT time  | SUBJECT(patid) COVTYPE(UN) .
```

- The procedure fits a model for *weight*. You believe that there is a linear relationship between weight and time, but there are also patient-specific effects on the size of the relationship that can be accounted for by random effects.
- The fixed-effects portion of the model includes *time*.
- The random-effects portion of the model includes the intercept and *time* with subjects identified by *patid* and an unstructured covariance matrix structure. You are, in essence, fitting a linear regression to each patient in which the regression coefficients are random effects; hence it is a “random coefficients” model.
- The `PRINT` subcommand requests the estimates of fixed effects, the random effects covariance (G) matrix, and tests of the covariance parameters.
- All other options are set to their default values.

Model 6: Multilevel Analysis

Suppose that *SCORE* is the score of a particular achievement test given over *TIME*. *STUDENT* is nested within *CLASS*, and *CLASS* is nested within *SCHOOL*.

```
MIXED SCORE WITH TIME
  /FIXED = TIME
  /RANDOM = INTERCEPT TIME | SUBJECT(SCHOOL) COVTYPE(ID)
  /RANDOM = INTERCEPT TIME | SUBJECT(SCHOOL*CLASS) COVTYPE(ID)
  /RANDOM = INTERCEPT TIME | SUBJECT(SCHOOL*CLASS*STUDENT) COVTYPE(ID).
```

Model 7: Unconditional Linear Growth Model

Suppose that *SUBJ* is the individual's identification and *Y* is the response of an individual observed over *TIME*. The covariance structure is unspecified.

```
MIXED Y WITH TIME
  /FIXED = TIME
  /RANDOM = INTERCEPT TIME | SUBJECT(SUBJ) COVTYPE(ID).
```

Model 8: Linear Growth Model with a Person-Level Covariate

Suppose that *PCOVAR* is the person-level covariate.

```
MIXED Y WITH TIME PCOVAR
  /FIXED = TIME PCOVAR TIME*PCOVAR
  /RANDOM = INTERCEPT TIME | SUBJECT(SUBJ) COVTYPE(ID).
```

Model 9: Repeated Measures Analysis

Suppose that *SUBJ* is the individual's identification and *Y* is the response of an individual observed over several *STAGE*s. The covariance structure is compound symmetry.

```
MIXED Y BY STAGE
  /FIXED = STAGE
  /REPEATED = STAGE | SUBJECT(SUBJ) COVTYPE(CS).
```

Model 10: Repeated Measures Analysis with Time-Dependent Covariate

Suppose that *SUBJ* is the individual's identification and *Y* is the response of an individual observed over several *STAGE*s. *X* is an individual-level covariate that also measures over several *STAGE*s. The residual covariance matrix structure is *AR(1)*.

```
MIXED Y BY STAGE WITH X
  /FIXED = X STAGE
  /REPEATED = STAGE | SUBJECT(SUBJ) COVTYPE(AR1).
```

Model 11: Repeated Measures and Random Effects

```
MIXED
  sales  BY marketid mktsize promo locid
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(10) SCORING(1)
  SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
  PCONVERGE(0.000001, ABSOLUTE)
  /FIXED = promo  | SSTYPE(3)
  /METHOD = REML
```

```
  /PRINT = G R SOLUTION TESTCOV
  /RANDOM = INTERCEPT | SUBJECT(marketid) COVTYPE(ID)
  /REPEATED = week  | SUBJECT(locid marketid ) COVTYPE(AR1) .
```

- The procedure fits a model for *sales*. The factor of interest is *promo* (the promotion offered); other factors are used to account for location and market effects.
- The fixed-effects portion of the model includes *promo*.
- The random-effects portion of the model includes the intercept with "subjects" identified by *marketid* and a scaled identity covariance matrix structure. This is equivalent to, but more computationally efficient than, specifying `/RANDOM marketid | COVTYPE(ID)`.
- The repeated-effects portion of the model includes *week* with subjects identified by the unique combinations of *locid* and *marketid*, and a first order autoregressive covariance matrix structure.
- The `PRINT` subcommand requests the estimates of fixed effects, the random effects covariance (G) matrix, the residual covariance (R) matrix, and tests of the covariance parameters.
- All other options are set to their default values.

Model 12: Crossover Trial

A crossover trial is a study where subjects are exposed to a series of treatments. The end goal is to establish treatment effects, but the model must also account for the effects of the sequencing of treatments. There are a number of ways to specify the model; for example, by using a random effects model of the customers within sequence:

```
MIXED
  amtspent  BY coupval seq carry week custid
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(10) SCORING(1)
  SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
  PCONVERGE(0.000001, ABSOLUTE)
  /FIXED = coupval seq carry week  | SSTYPE(3)
  /METHOD = REML
  /PRINT = G SOLUTION TESTCOV
 /RANDOM = custid(seq) | COVTYPE(ID).
```

- The procedure fits a model for a carryover trial for *amtspent* (the amount spent by the customer). The factor of interest is *coupval* (the value of the offer to the customer); other factors are used to account for the effects of the sequencing of offers.
- The fixed-effects portion of the model includes *coupval*, *seq* (the sequence in which the customer received offers), *carry* (an indicator to account for carryover effects), and *week* (the week of the trial in which the observation is taken).
- The random-effects portion of the model includes *custid(seq)* (customers within sequence) with a scaled identity covariance matrix structure.
- The `PRINT` subcommand requests the estimates of fixed effects, the random effects covariance (G) matrix, and tests of the covariance parameters.
- All other options are set to their default values.

You can also use a repeated effects model on the time periods:

```
MIXED
  amtspent  BY coupval carry week seq
```

```
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(100) SCORING(1)
  SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
  PCONVERGE(0.000001, ABSOLUTE)
  /FIXED = coupval seq carry week | SSTYPE(3)
  /METHOD = REML
  /PRINT = R SOLUTION TESTCOV
  /REPEATED = week  | SUBJECT(custid ) COVTYPE(AR1) .
```

- The procedure fits a model for a carryover trial for *amtspent* (the amount spent by the customer). The factor of interest is *coupval* (the value of the offer to the customer); other factors are used to account for the effects of the sequencing of offers.
- The fixed-effects portion of the model includes *coupval*, *seq* (the sequence in which the customer received offers), *carry* (an indicator to account for carryover effects), and *week* (the week of the trial in which the observation is taken).
- The repeated-effects portion of the model includes *week* with subjects identified by *custid* and a first-order autoregressive covariance matrix structure.
- The `PRINT` subcommand requests the estimates of fixed effects, the residual covariance (R) matrix, and tests of the covariance parameters.
- All other options are set to their default values.

Lastly, you can use a repeated effects model on the treatments:

```
MIXED
  amtspent  BY coupval carry week seq
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(10) SCORING(1)
  SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
  PCONVERGE(0.000001, ABSOLUTE)
  /FIXED = coupval seq carry week | SSTYPE(3)
  /METHOD = REML
  /PRINT = R SOLUTION TESTCOV
  /REPEATED = coupval  | SUBJECT(custid ) COVTYPE(UN) .
```

- The procedure fits a model for a carryover trial for *amtspent* (the amount spent by the customer). The factor of interest is *coupval* (the value of the offer to the customer); other factors are used to account for the effects of the sequencing of offers.
- The fixed-effects portion of the model includes *coupval*, *seq* (the sequence in which the customer received offers), *carry* (an indicator to account for carryover effects), and *week* (the week of the trial in which the observation is taken).
- The repeated-effects portion of the model includes *coupval* with subjects identified by *custid* and an unstructured covariance matrix structure.
- The `PRINT` subcommand requests the estimates of fixed effects, the residual covariance (R) matrix, and tests of the covariance parameters.
- All other options are set to their default values.

Case Frequency

- If a `WEIGHT` variable is specified, its values are used as frequency weights by the `MIXED` procedure.

- Cases with missing weights or weights less than 0.5 are not used in the analyses.
- The weight values are rounded to the nearest whole numbers before use. For example, 0.5 is rounded to 1, and 2.4 is rounded to 2.

Covariance Structure List

The following is the list of covariance structures being offered by the `MIXED` procedure. Unless otherwise implied or stated, the structures are not constrained to be non-negative definite in order to avoid nonlinear constraints and to reduce the optimization complexity. However, the variances are restricted to be non-negative.

- Separate covariance matrices are computed for each random effect; that is, while levels of a given random effect are allowed to co-vary, they are considered independent of the levels of other random effects.

AD1 *First-order ante-dependence.* The constraint $|\rho_k| \leq 1$ is imposed for stationarity.

$$\begin{bmatrix} \sigma_1^2 & \sigma_2\sigma_1\rho_1 & \sigma_3\sigma_1\rho_1\rho_2 & \sigma_4\sigma_1\rho_1\rho_2\rho_3 \\ \sigma_2\sigma_1\rho_1 & \sigma_2^2 & \sigma_3\sigma_2\rho_2 & \sigma_4\sigma_2\rho_2\rho_3 \\ \sigma_3\sigma_1\rho_1\rho_2 & \sigma_3\sigma_2\rho_2 & \sigma_3^2 & \sigma_4\sigma_3\rho_3 \\ \sigma_4\sigma_1\rho_1\rho_2\rho_3 & \sigma_4\sigma_2\rho_2\rho_3 & \sigma_4\sigma_3\rho_3 & \sigma_4^2 \end{bmatrix}$$

AR1 *First-order autoregressive.* The constraint $|\rho| \leq 1$ is imposed for stationarity.

$$\sigma^2 \begin{bmatrix} 1 & \rho & \rho^2 & \rho^3 \\ \rho & 1 & \rho & \rho^2 \\ \rho^2 & \rho & 1 & \rho \\ \rho^3 & \rho^2 & \rho & 1 \end{bmatrix}$$

ARH1 *Heterogenous first-order autoregressive.* The constraint $|\rho_k| \leq 1$ is imposed for stationarity.

$$\begin{bmatrix} \sigma_1^2 & \sigma_2\sigma_1\rho & \sigma_3\sigma_1\rho^2 & \sigma_4\sigma_1\rho^3 \\ \sigma_2\sigma_1\rho & \sigma_2^2 & \sigma_3\sigma_2\rho & \sigma_4\sigma_2\rho^2 \\ \sigma_3\sigma_1\rho^2 & \sigma_3\sigma_2\rho & \sigma_3^2 & \sigma_4\sigma_3\rho \\ \sigma_4\sigma_1\rho^3 & \sigma_4\sigma_2\rho^2 & \sigma_4\sigma_3\rho & \sigma_4^2 \end{bmatrix}$$

ARMA11 *Autoregressive moving average (1,1).* The constraints $|\varphi| \leq 1$ and $|\rho| \leq 1$ are imposed for stationarity.

$$\sigma^2 \begin{bmatrix} 1 & \phi\rho & \phi\rho^2 & \phi\rho^3 \\ \phi\rho & 1 & \phi\rho & \phi\rho^2 \\ \phi\rho^2 & \phi\rho & 1 & \phi\rho \\ \phi\rho^3 & \phi\rho^2 & \phi\rho & 1 \end{bmatrix}$$

CS *Compound symmetry.* This structure has constant variance and constant covariance.

$$\begin{bmatrix} \sigma^2+\sigma_1 & \sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma^2+\sigma_1 & \sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma^2+\sigma_1 & \sigma_1 \\ \sigma_1 & \sigma_1 & \sigma_1 & \sigma^2+\sigma_1 \end{bmatrix}$$

CSH *Heterogenous compound symmetry.* This structure has non-constant variance and constant correlation.

$$\begin{bmatrix} \sigma_1^2 & \sigma_2\sigma_1\rho & \sigma_3\sigma_1\rho & \sigma_4\sigma_1\rho \\ \sigma_2\sigma_1\rho & \sigma_2^2 & \sigma_3\sigma_2\rho & \sigma_4\sigma_2\rho \\ \sigma_3\sigma_1\rho & \sigma_3\sigma_2\rho & \sigma_3^2 & \sigma_4\sigma_3\rho \\ \sigma_4\sigma_1\rho & \sigma_4\sigma_2\rho & \sigma_4\sigma_3\rho & \sigma_4^2 \end{bmatrix}$$

CSR *Compound symmetry with correlation parameterization.* This structure has constant variance and constant covariance.

$$\sigma^2 \begin{bmatrix} 1 & \rho & \rho & \rho \\ \rho & 1 & \rho & \rho \\ \rho & \rho & 1 & \rho \\ \rho & \rho & \rho & 1 \end{bmatrix}$$

DIAG *Diagonal.* This is a diagonal structure with heterogenous variance. This is the default covariance structure for repeated effects.

$$\begin{bmatrix} \sigma_1^2 & 0 & 0 & 0 \\ 0 & \sigma_2^2 & 0 & 0 \\ 0 & 0 & \sigma_3^2 & 0 \\ 0 & 0 & 0 & \sigma_4^2 \end{bmatrix}$$

FA1 *First-order factor analytic with constant diagonal offset* ($d \geq 0$).

$$\begin{bmatrix} \lambda_1^2 + d & \lambda_2\lambda_1 & \lambda_3\lambda_1 & \lambda_4\lambda_1 \\ \lambda_2\lambda_1 & \lambda_2^2 + d & \lambda_3\lambda_2 & \lambda_4\lambda_2 \\ \lambda_3\lambda_1 & \lambda_3\lambda_2 & \lambda_3^2 + d & \lambda_4\lambda_3 \\ \lambda_4\lambda_1 & \lambda_4\lambda_2 & \lambda_4\lambda_3 & \lambda_4^2 + d \end{bmatrix}$$

FAH1 *First-order factor analytic with heterogenous diagonal offsets* ($d_k \geq 0$).

$$\begin{bmatrix} \lambda_1^2 + d_1 & \lambda_2\lambda_1 & \lambda_3\lambda_1 & \lambda_4\lambda_1 \\ \lambda_2\lambda_1 & \lambda_2^2 + d_2 & \lambda_3\lambda_2 & \lambda_4\lambda_2 \\ \lambda_3\lambda_1 & \lambda_3\lambda_2 & \lambda_3^2 + d_3 & \lambda_4\lambda_3 \\ \lambda_4\lambda_1 & \lambda_4\lambda_2 & \lambda_4\lambda_3 & \lambda_4^2 + d_4 \end{bmatrix}$$

HF *Huynh-Feldt.* This is a circular matrix that satisfies the condition $\sigma_i^2 + \sigma_j^2 - 2\sigma_{ij} = 2\lambda$.

$$\begin{bmatrix} \sigma_1^2 & \frac{\sigma_1^2+\sigma_2^2}{2} - \lambda & \frac{\sigma_1^2+\sigma_3^2}{2} - \lambda & \frac{\sigma_1^2+\sigma_4^2}{2} - \lambda \\ \frac{\sigma_1^2+\sigma_2^2}{2} - \lambda & \sigma_2^2 & \frac{\sigma_2^2+\sigma_3^2}{2} - \lambda & \frac{\sigma_2^2+\sigma_4^2}{2} - \lambda \\ \frac{\sigma_1^2+\sigma_3^2}{2} - \lambda & \frac{\sigma_2^2+\sigma_3^2}{2} - \lambda & \sigma_3^2 & \frac{\sigma_3^2+\sigma_4^2}{2} - \lambda \\ \frac{\sigma_1^2+\sigma_4^2}{2} - \lambda & \frac{\sigma_2^2+\sigma_4^2}{2} - \lambda & \frac{\sigma_3^2+\sigma_4^2}{2} - \lambda & \sigma_4^2 \end{bmatrix}$$

ID *Identity.* This is a scaled identity matrix.

$$\sigma^2 \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

TP *Toeplitz* ($|\rho_k| \leq 1$).

$$\sigma^2 \begin{bmatrix} 1 & \rho_1 & \rho_2 & \rho_3 \\ \rho_1 & 1 & \rho_1 & \rho_2 \\ \rho_2 & \rho_1 & 1 & \rho_1 \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix}$$

TPH *Heterogenous Toeplitz* ($|\rho_k| \leq 1$).

$$\begin{bmatrix} \sigma_1^2 & \sigma_2\sigma_1\rho_1 & \sigma_3\sigma_1\rho_2 & \sigma_4\sigma_1\rho_3 \\ \sigma_2\sigma_1\rho_1 & \sigma_2^2 & \sigma_3\sigma_2\rho_1 & \sigma_4\sigma_2\rho_2 \\ \sigma_3\sigma_1\rho_2 & \sigma_3\sigma_2\rho_1 & \sigma_3^2 & \sigma_4\sigma_3\rho_1 \\ \sigma_4\sigma_1\rho_3 & \sigma_4\sigma_2\rho_2 & \sigma_4\sigma_3\rho_1 & \sigma_4^2 \end{bmatrix}$$

UN *Unstructured.* This is a completely general covariance matrix.

$$\begin{bmatrix} \sigma_1^2 & \sigma_{21} & \sigma_{31} & \sigma_{41} \\ \sigma_{21} & \sigma_2^2 & \sigma_{32} & \sigma_{42} \\ \sigma_{31} & \sigma_{32} & \sigma_3^2 & \sigma_{43} \\ \sigma_{41} & \sigma_{42} & \sigma_{43} & \sigma_4^2 \end{bmatrix}$$

UNR *Unstructured correlations (*$|\rho_{jk}| \leq 1$).

$$\begin{bmatrix} \sigma_1^2 & \sigma_2\sigma_1\rho_{21} & \sigma_3\sigma_1\rho_{31} & \sigma_4\sigma_1\rho_{41} \\ \sigma_2\sigma_1\rho_{21} & \sigma_2^2 & \sigma_3\sigma_2\rho_{32} & \sigma_4\sigma_2\rho_{42} \\ \sigma_3\sigma_1\rho_{31} & \sigma_3\sigma_2\rho_{32} & \sigma_3^2 & \sigma_4\sigma_3\rho_{43} \\ \sigma_4\sigma_1\rho_{41} & \sigma_4\sigma_2\rho_{42} & \sigma_4\sigma_3\rho_{43} & \sigma_4^2 \end{bmatrix}$$

VC *Variance components.* This is the default covariance structure for random effects. When the variance components structure is specified on a `RANDOM` subcommand, a scaled identity (`ID`) structure is assigned to each of the effects specified on the subcommand. If the variance components structure is specified on the `REPEATED` subcommand, it is replaced by the diagonal (`DIAG`) structure.

Variable List

The variable list specifies the dependent variable, the factors, and the covariates in the model.

- The dependent variable must be the first specification on `MIXED`.
- The names of the factors, if any, must be preceded by the keyword `BY`.
- The names of the covariates, if any, must be preceded by the keyword `WITH`.
- The dependent variable and the covariates must be numeric.
- The factor variables can be of any type (numeric and string).
- Only cases with no missing values in all of the variables specified will be used.

CRITERIA Subcommand

The `CRITERIA` subcommand controls the iterative algorithm used in the estimation and specifies numerical tolerance for checking singularity.

CIN(value)	*Confidence interval level.* This value is used whenever a confidence interval is constructed. Specify a value greater than or equal to 0 and less than 100. The default value is 95.
HCONVERGE(value, type)	*Hessian convergence criterion.* Convergence is assumed if $g'_k H_k^{-1} g_k$ is less than a multiplier of *value*. The multiplier is 1 for `ABSOLUTE` type and is the absolute value of the current log-likelihood function for `RELATIVE` type. The criterion is not used if *value* equals 0. This criterion is not used by default. Specify a non-negative value and a measure type of convergence.
LCONVERGE(value, type)	*Log-likelihood function convergence criterion.* Convergence is assumed if the `ABSOLUTE` or `RELATIVE` change in the log-likelihood function is less than *value*. The criterion is not used if *a* equals 0. This criterion is not used by default. Specify a non-negative value and a measure type of convergence.
MXITER(n)	*Maximum number of iterations.* Specify a non-negative integer. The default value is 100.
PCONVERGE(value, type)	*Parameter estimates convergence criterion.* Convergence is assumed if the maximum `ABSOLUTE` or maximum `RELATIVE` change in the parameter estimates is less than *value*. The criterion is not used if *a* equals 0. Specify a non-negative value and a measure type of convergence. The default value for *a* is 10^{-6}.
MXSTEP(n)	*Maximum step-halving allowed.* At each iteration, the step size is reduced by a factor of 0.5 until the log-likelihood increases or maximum step-halving is reached. Specify a positive integer. The default value is 10.
SCORING(n)	*Apply scoring algorithm.* Requests to use the Fisher scoring algorithm up to iteration number *n*. Specify a positive integer. The default is 1.
SINGULAR(value)	*Value used as tolerance in checking singularity.* Specify a positive value. The default value is 10^{-12}.

Example

```
MIXED SCORE BY SCHOOL CLASS WITH AGE
/CRITERIA = CIN(90) LCONVERGE(0) MXITER(50) PCONVERGE(1E-5 RELATIVE)
/FIXED = AGE
/RANDOM = SCHOOL CLASS.
```

- The `CRITERIA` subcommand requests that a 90% confidence interval be calculated whenever appropriate.
- The log-likelihood convergence criterion is not used. Convergence is attained when the maximum relative change in parameter estimates is less than 0.00001 and number of iterations is less than 50.

Example

```
MIXED SCORE BY SCHOOL CLASS WITH AGE
  /CRITERIA = MXITER(100) SCORING(100)
  /FIXED = AGE
```

```
/RANDOM = SCHOOL CLASS.
```

- The Fisher scoring algorithm is used for all iterations.

EMMEANS Subcommand

EMMEANS displays estimated marginal means of the dependent variable in the cells and their standard errors for the specified factors. Note that these are predicted, not observed, means.

- The TABLES keyword, followed by an option in parentheses, is required. COMPARE is optional; if specified, it must follow TABLES.
- Multiple EMMEANS subcommands are allowed. Each is treated independently.
- If identical EMMEANS subcommands are specified, only the last identical subcommand is in effect. EMMEANS subcommands that are redundant but not identical (for example, crossed factor combinations such as *A*B* and *B*A*) are all processed.

TABLES(option) *Table specification.* Valid options are the keyword OVERALL, factors appearing on the factor list, and crossed factors constructed of factors on the factor list. Crossed factors can be specified by using an asterisk (*) or the keyword BY. All factors in a crossed factor specification must be unique.

If OVERALL is specified, the estimated marginal means of the dependent variable are displayed, collapsing over all factors.

If a factor, or a crossing factor, is specified on the TABLES keyword, MIXED will compute the estimated marginal mean for each level combination of the specified factor(s), collapsing over all other factors not specified with TABLES.

WITH (option) *Covariate values.* Valid options are covariates appearing on the covariate list on the VARIABLES subcommand. Each covariate must be followed by a numeric value or the keyword MEAN.

If a numeric value is used, the estimated marginal mean will be computed by holding the specified covariate at the supplied value.

When the keyword MEAN is used, the estimated marginal mean will be computed by holding the covariate at its overall mean. If a covariate is not specified in the WITH option, its overall mean will be used in estimated marginal mean calculations.

COMPARE(factor)
REFCAT(value)
ADJ(method) *Main- or simple-main-effects omnibus tests and pairwise comparisons of the dependent variable.* This option gives the mean difference, standard error, degrees of freedom, significance, and confidence intervals for each pair of levels for the effect specified in the COMPARE keyword, and an omnibus test for that effect. If only one factor is specified on TABLES, COMPARE can be specified by itself; otherwise, the factor specification is required. In this case, levels of the specified factor are compared with each other for each level of the other factors in the interaction.

The optional ADJ keyword allows you to apply an adjustment to the confidence intervals and significance values to account for multiple comparisons. Methods available are LSD (no adjustment), BONFERRONI, or SIDAK.

By default, all pairwise comparisons of the specified factor will be constructed. Optionally, comparisons can be made to a reference category by specifying the value of that category after the REFCAT keyword. If the compare factor is a string variable, the category value must be a quoted string. If the compare factor is a numeric variable, the category value should be specified as an unquoted numeric value. Alternatively, the keywords FIRST or LAST can be used to specify whether the first or the last category will be used as a reference category.

Example

```
MIXED Y BY A B WITH X
  /FIXED A B X
  /EMMEANS TABLES(A*B) WITH(X=0.23) COMPARE(A) ADJ(SIDAK)
  /EMMEANS TABLES(A*B) WITH(X=MEAN) COMPARE(A) REFCAT(LAST) ADJ(LSD).
```

- In the example, the first EMMEANS subcommand will compute estimated marginal means for all level combinations of *A*B* by fixing the covariate *X* at 0.23. Then for each level of *B*, all pairwise comparisons on *A* will be performed using SIDAK adjustment.
- In the second EMMEANS subcommand, the estimated marginal means will be computed by fixing the covariate *X* at its mean. Since REFCAT(LAST) is specified, comparison will be made to the last category of factor *A* using LSD adjustment.

FIXED Subcommand

The FIXED subcommand specifies the fixed effects in the mixed model.

- Specify a list of terms to be included in the model, separated by commas or spaces.
- The intercept term is included by default.
- The default model is generated if the FIXED subcommand is omitted or empty. The default model consists of only the intercept term (if included).
- To explicitly include the intercept term, specify the keyword INTERCEPT on the FIXED subcommand. The INTERCEPT term must be specified first on the FIXED subcommand.
- To include a main-effect term, enter the name of the factor on the FIXED subcommand.
- To include an interaction-effect term among factors, use the keyword BY or the asterisk (*) to connect factors involved in the interaction. For example, A*B*C means a three-way interaction effect of the factors *A, B,* and *C*. The expression A BY B BY C is equivalent to A*B*C. Factors inside an interaction effect must be distinct. Expressions such as A*C*A and A*A are invalid.
- To include a nested-effect term, use the keyword WITHIN or a pair of parentheses on the FIXED subcommand. For example, A(B) means that *A* is nested within *B*, where *A* and *B* are factors. The expression A WITHIN B is equivalent to A(B). Factors inside a nested effect must be distinct. Expressions such as A(A) and A(B*A) are invalid.

- Multiple-level nesting is supported. For example, `A(B(C))` means that *B* is nested within *C*, and *A* is nested within *B(C)*. When more than one pair of parentheses is present, each pair of parentheses must be enclosed or nested within another pair of parentheses. Thus, `A(B)(C)` is invalid.
- Nesting within an interaction effect is valid. For example, `A(B*C)` means that *A* is nested within *B*C*.
- Interactions among nested effects are allowed. The correct syntax is the interaction followed by the common nested effect inside the parentheses. For example, the interaction between *A* and *B* within levels of *C* should be specified as `A*B(C)` instead of `A(C)*B(C)`.
- To include a covariate term in the model, enter the name of the covariate on the `FIXED` subcommand.
- Covariates can be connected using the keyword `BY` or the asterisk (*). For example, `X*X` is the product of *X* and itself. This is equivalent to entering a covariate whose values are the squared values of *X*.
- Factor and covariate effects can be connected in many ways. Suppose that *A* and *B* are factors and *X* and *Y* are covariates. Examples of valid combinations of factor and covariate effects are `A*X`, `A*B*X`, `X(A)`, `X(A*B)`, `X*A(B)`, `X*Y(A*B)`, and `A*B*X*Y`.
- No effects can be nested within a covariate effect. Suppose that *A* and *B* are factors and *X* and *Y* are covariates. The effects `A(X)`, `A(B*Y)`, `X(Y)`, and `X(B*Y)` are invalid.
- The following options, which are specific for the fixed effects, can be entered after the effects. Use the vertical bar (|) to precede the options.

NOINT *No intercept.* The intercept terms are excluded from the fixed effects.

SSTYPE(n) *Type of sum of squares.* Specify the methods for partitioning the sums of squares. Specify $n = 1$ for Type I sum of squares or $n = 3$ for Type III sum of squares. The default is Type III sum of squares.

Example

```
MIXED  SCORE BY SCHOOL CLASS WITH AGE PRETEST
  /FIXED = AGE(SCHOOL) AGE*PRETEST(SCHOOL)
  /RANDOM = CLASS.
```

- In this example, the fixed-effects design consists of the default `INTERCEPT`, a nested effect *AGE* within *SCHOOL*, and another nested effect of the product of *AGE* and *PRETEST* within *SCHOOL*.

Example

```
MIXED SCORE BY SCHOOL CLASS
  /FIXED = | NOINT
  /RANDOM = SCHOOL CLASS.
```

- In this example, a purely random-effects model is fitted. The random effects are *SCHOOL* and *CLASS*. The fixed-effects design is empty because the implicit intercept term is removed by the `NOINT` keyword.
- You can explicitly insert the `INTERCEPT` effect as `/FIXED = INTERCEPT | NOINT`. But the specification will be identical to `/FIXED = | NOINT`.

METHOD Subcommand

The METHOD subcommand specifies the estimation method.

- If this subcommand is not specified, the default is REML.
- The keywords ML and REML are mutually exclusive. Only one of them can be specified once.

ML *Maximum likelihood.*

REML *Restricted maximum likelihood.* This is the default.

MISSING Subcommand

The MISSING subcommand specifies the way to handle cases with user-missing values.

- If this subcommand is not specified, the default is EXCLUDE.
- Cases, which contain system-missing values in one of the variables, are always deleted.
- The keywords EXCLUDE and INCLUDE are mutually exclusive. Only one of them can be specified at once.

EXCLUDE *Exclude both user-missing and system-missing values.* This is the default.

INCLUDE *User-missing values are treated as valid.* System-missing values cannot be included in the analysis.

PRINT Subcommand

The PRINT subcommand specifies additional output. If no PRINT subcommand is specified, the default output includes:

- A model dimension summary table
- A covariance parameter estimates table
- A model fit summary table
- A test of fixed effects table

CORB *Asymptotic correlation matrix of the fixed-effects parameter estimates.*

COVB *Asymptotic covariance matrix of the fixed-effects parameter estimates.*

CPS *Case processing summary.* Displays the sorted values of the factors, the repeated measure variables, the repeated measure subjects, the random-effects subjects, and their frequencies.

DESCRIPTIVES *Descriptive statistics.* Displays the sample sizes, the means, and the standard deviations of the dependent variable, and covariates (if specified). These statistics are displayed for each distinct combination of the factors.

G *Estimated covariance matrix of random effects.* This keyword is accepted only when at least one RANDOM subcommand is specified. Otherwise, it will be ignored. If a SUBJECT variable is specified for a random effect, then the common block is displayed.

HISTORY(n) *Iteration history.* The table contains the log-likelihood function value and parameter estimates for every *n* iterations beginning with the 0*th* iteration (the initial estimates). The default is to print every iteration (n = 1). If HISTORY is specified, the last iteration is always printed regardless of the value of *n*.

LMATRIX	*Estimable functions.* Displays the estimable functions used for testing the fixed effects and for testing the custom hypothesis.
R	*Estimated covariance matrix of residual.* This keyword is accepted only when a REPEATED subcommand is specified. Otherwise, it will be ignored. If a SUBJECT variable is specified, the common block is displayed.
SOLUTION	*A solution for the fixed-effects and the random-effects parameters.* The fixed-effects and the random-effects parameter estimates are displayed. Their approximate standard errors are also displayed.
TESTCOV	*Tests for the covariance parameters.* Displays the asymptotic standard errors and Wald tests for the covariance parameters.

RANDOM Subcommand

The RANDOM subcommand specifies the random effects in the mixed model.

- Depending on the covariance type specified, random effects specified in one RANDOM subcommand may be correlated.
- One covariance G matrix will be constructed for each RANDOM subcommand. The dimension of the random effect covariance G matrix is equal to the sum of the levels of all random effects in the subcommand.
- When the variance components (VC) structure is specified, a scaled identity (ID) structure will be assigned to each of the effects specified. This is the default covariance type for the RANDOM subcommand.
- Note that the RANDOM subcommand in the MIXED procedure is different in syntax from the RANDOM subcommand in the GLM and VARCOMP procedures.
- Use a separate RANDOM subcommand when a different covariance structure is assumed for a list of random effects. If the same effect is listed on more than one RANDOM subcommand, it must be associated with a different SUBJECT combination.
- Specify a list of terms to be included in the model, separated by commas or spaces.
- No random effects are included in the mixed model unless a RANDOM subcommand is specified correctly.
- Specify the keyword INTERCEPT to include the intercept as a random effect. The MIXED procedure does not include the intercept in the RANDOM subcommand by default. The INTERCEPT term must be specified first on the RANDOM subcommand.
- To include a main-effect term, enter the name of the factor on the RANDOM subcommand.
- To include an interaction-effect term among factors, use the keyword BY or the asterisk (*) to join factors involved in the interaction. For example, A*B*C means a three-way interaction effect of *A*, *B*, and *C*, where *A*, *B*, and *C* are factors. The expression A BY B BY C is equivalent to A*B*C. Factors inside an interaction effect must be distinct. Expressions such as A*C*A and A*A are invalid.
- To include a nested-effect term, use the keyword WITHIN or a pair of parentheses on the RANDOM subcommand. For example, A(B) means that *A* is nested within *B*, where *A* and *B* are factors. The expression A WITHIN B is equivalent to A(B). Factors inside a nested effect must be distinct. Expressions such as A(A) and A(B*A) are invalid.

- Multiple-level nesting is supported. For example, A(B(C)) means that *B* is nested within *C*, and *A* is nested within *B(C)*. When more than one pair of parentheses is present, each pair of parentheses must be enclosed or nested within another pair of parentheses. Thus, A(B)(C) is invalid.
- Nesting within an interaction effect is valid. For example, A(B*C) means that *A* is nested within *B***C*.
- Interactions among nested effects are allowed. The correct syntax is the interaction followed by the common nested effect inside the parentheses. For example, the interaction between *A* and *B* within levels of *C* should be specified as A*B(C) instead of A(C)*B(C).
- To include a covariate term in the model, enter the name of the covariate on the FIXED subcommand.
- Covariates can be connected using the keyword BY or the asterisk (*). For example, X*X is the product of *X* and itself. This is equivalent to entering a covariate whose values are the squared values of *X*.
- Factor and covariate effects can be connected in many ways. Suppose that *A* and *B* are factors and *X* and *Y* are covariates. Examples of valid combinations of factor and covariate effects are A*X, A*B*X, X(A), X(A*B), X*A(B), X*Y(A*B), and A*B*X*Y.
- No effects can be nested within a covariate effect. Suppose that *A* and *B* are factors and *X* and *Y* are covariates. The effects A(X), A(B*Y), X(Y), and X(B*Y) are invalid.
- The following options, which are specific for the random effects, can be entered after the effects. Use the vertical bar (|) to precede the options.

SUBJECT(varname*varname*…) *Identify the subjects.* Complete independence is assumed across subjects, thus producing a block-diagonal structure in the covariance matrix of the random effect with identical blocks. Specify a list of variable names (of any type) connected by asterisks. The number of subjects is equal to the number of distinct combinations of values of the variables. A case will not be used if it contains a missing value on any of the subject variables.

COVTYPE(type) *Covariance structure.* Specify the covariance structure of the identical blocks for the random effects (see Covariance Structure List on p. 1248). The default covariance structure for random effects is VC.

- If the REPEATED subcommand is specified, the variables in the RANDOM subject list must be a subset of the variables in the REPEATED subject list.
- Random effects are considered independent of each other, and a separate covariance matrix is computed for each effect.

Example

```
MIXED  SCORE BY SCHOOL CLASS
  /RANDOM = INTERCEPT SCHOOL CLASS.
```

REGWGT Subcommand

The REGWGT subcommand specifies the name of a variable containing the regression weights.

- Specify a numeric variable name following the REGWGT subcommand.
- Cases with missing or non-positive weights are not used in the analyses.
- The regression weights will be applied only to the covariance matrix of the residual term.

REPEATED Subcommand

The REPEATED subcommand specifies the residual covariance matrix in the mixed-effects model. If no REPEATED subcommand is specified, the residual covariance matrix assumes the form of a scaled identity matrix with the scale being the usual residual variance.

- Specify a list of variable names (of any type) connected by asterisks (repeated measure) following the REPEATED subcommand.
- Distinct combinations of values of the variables are used simply to identify the repeated observations. Order of the values will determine the order of occurrence of the repeated observations. Therefore, the lowest values of the variables associate with the first repeated observation, and the highest values associate with the last repeated observation.
- The VC covariance structure is obsolete in the REPEATED subcommand. If it is specified, it will be replaced with the DIAG covariance structure. An annotation will be made in the output to indicate this change.
- The default covariance type for repeated effects is DIAG.
- The following keywords, which are specific for the REPEATED subcommand, can be entered after the effects. Use the vertical bar (|) to precede the options.

SUBJECT(varname*varname*…)	*Identify the subjects.* Complete independence is assumed across subjects, thus producing a block-diagonal structure in the residual covariance matrix with identical blocks. The number of subjects is equal to the number of distinct combinations of values of the variables. A case will not be used if it contains a missing value on any of the subject variables.
COVTYPE(type)	*Covariance structure.* Specify the covariance structure of the identical blocks for the residual covariance matrix (see Covariance Structure List on p. 1248). The default structure for repeated effects is DIAG.

- The SUBJECT keyword must be specified to identify the subjects in a repeated measurement analysis. The analysis will not be performed if this keyword is omitted.
- The list of subject variables must contain all of the subject variables specified in all RANDOM subcommands.
- Any variable used in the repeated measure list must not be used in the repeated subject specification.

Example

```
MIXED  SCORE BY CLASS
  /RANDOM = CLASS | SUBJECT(SCHOOL)
  /REPEATED = FLOOR | SUBJECT(SCHOOL*STUDENT).
```

However, the syntax in each of the following examples is invalid:

```
MIXED  SCORE BY CLASS
  /RANDOM = CLASS | SUBJECT(SCHOOL)
  /REPEATED = FLOOR | SUBJECT(STUDENT).
```

```
MIXED  SCORE BY CLASS
  /RANDOM = CLASS | SUBJECT(SCHOOL*STUDENT)
  /REPEATED = FLOOR | SUBJECT(STUDENT).
```

```
MIXED  SCORE BY CLASS
  /RANDOM = CLASS | SUBJECT(SCHOOL)
  /REPEATED = STUDENT | SUBJECT(STUDENT*SCHOOL).
```

- In the first two examples, the RANDOM subject list contains a variable not on the REPEATED subject list.
- In the third example, the REPEATED subject list contains a variable on the REPEATED variable list.

SAVE Subcommand

Use the SAVE subcommand to save one or more casewise statistics to the active dataset.

- Specify one or more temporary variables, each followed by an optional new name in parentheses.
- If new names are not specified, default names are generated.

FIXPRED	*Fixed predicted values.* The regression means without the random effects.
PRED	*Predicted values.* The model fitted value.
RESID	*Residuals.* The data value minus the predicted value.
SEFIXP	*Standard error of fixed predicted values.* These are the standard error estimates for the fixed effects predicted values obtained by the keyword FIXPRED.
SEPRED	*Standard error of predicted values.* These are the standard error estimates for the overall predicted values obtained by the keyword PRED.
DFFIXP	*Degrees of freedom of fixed predicted values.* These are the Satterthwaite degrees of freedom for the fixed effects predicted values obtained by the keyword FIXPRED.
DFPRED	*Degrees of freedom of predicted values.* These are the Satterthwaite degrees of freedom for the fixed effects predicted values obtained by the keyword PRED.

Example

```
MIXED  SCORE BY SCHOOL CLASS WITH AGE
  /FIXED = AGE
  /RANDOM = SCHOOL CLASS(SCHOOL)
  /SAVE = FIXPRED(BLUE) PRED(BLUP) SEFIXP(SEBLUE) SEPRED(SEBLUP).
```

- The SAVE subcommand appends four variables to the active dataset: *BLUE*, containing the fixed predicted values, *BLUP*, containing the predicted values, *SEBLUE*, containing the standard error of *BLUE*, and *SEBLUP*, containing the standard error of *BLUP*.

TEST Subcommand

The TEST subcommand allows you to customize your hypotheses tests by directly specifying null hypotheses as linear combinations of parameters.

- Multiple TEST subcommands are allowed. Each is handled independently.
- The basic format for the TEST subcommand is an optional list of values enclosed in a pair of parentheses, an optional label in quotes, an effect name or the keyword ALL, and a list of values.
- When multiple linear combinations are specified within the same TEST subcommand, a semicolon (;) terminates each linear combination except the last one.
- At the end of a contrast coefficients row, you can use the option DIVISOR=value to specify a denominator for coefficients in that row. When specified, the contrast coefficients in that row will be divided by the given value. Note that the equals sign is required.
- The value list preceding the first effect or the keyword ALL contains the constants, to which the linear combinations are equated under the null hypotheses. If this value list is omitted, the constants are assumed to be zeros.
- The optional label is a string with a maximum length of 255 bytes. Only one label per TEST subcommand can be specified.
- The effect list is divided into two parts. The first part is for the fixed effects, and the second part is for the random effects. Both parts have the same syntax structure.
- Effects specified in the fixed-effect list should have already been specified or implied on the FIXED subcommand.
- Effects specified in the random-effect list should have already been specified on the RANDOM subcommand.
- To specify the coefficient for the intercept, use the keyword INTERCEPT. Only one value is expected to follow INTERCEPT.
- The number of values following an effect name must be equal to the number of parameters (including the redundant ones) corresponding to that effect. For example, if the effect *A**B* takes up to six parameters, then exactly six values must follow *A**B*.
- A number can be specified as a fraction with a positive denominator. For example, 1/3 or –1/3 are valid, but 1/–3 is invalid.
- When ALL is specified, only a list of values can follow. The number of values must be equal to the number of parameters (including the redundant ones) in the model.
- Effects appearing or implied on the FIXED and RANDOM subcommands but not specified on TEST are assumed to take the value 0 for all of their parameters.
- If ALL is specified for the first row in a TEST matrix, then all subsequent rows should begin with the ALL keyword.
- If effects are specified for the first row in a TEST matrix, then all subsequent rows should use the effect name (thus ALL is not allowed).
- When SUBJECT( ) is specified on a RANDOM subcommand, the coefficients given in the TEST subcommand will be divided by the number of subjects of that random effect automatically.

Example

```
MIXED Y BY A B C
  /FIX = A
  /RANDOM = B C
  /TEST = 'Contrasts of A' A 1/3 1/3 1/3; A 1 -1 0; A 1 -1/2 -1/2
  /TEST(1) = 'Contrast of B' | B 1 -1
  /TEST = 'BLUP at First Level of A'
          ALL 0 1 0 0 | 1 0 1 0;
          ALL         | 1 0 0 1;
          ALL 0 1 0 0;
          ALL 0 1 0 0 | 0 1 0 1.
```

Suppose that factor *A* has three levels and factors *B* and *C* each have two levels.

- The first `TEST` is labeled *Contrasts of A*. It performs three contrasts among levels of *A*. The first is technically not a contrast but the mean of level 1, level 2, and level 3 of *A*, the second is between level 1 and level 2 of *A*, and the third is between level 1 and the mean of level 2 and level 3 of *A*.
- The second `TEST` is labeled *Contrast of B*. Coefficients for *B* are preceded by the vertical bar (|) because *B* is a random effect. This contrast computes the difference between level 1 and level 2 of *B*, and tests if the difference equals 1.
- The third `TEST` is labeled *BLUP at First Level of A*. There are four parameters for the fixed effects (intercept and *A*), and there are four parameters for the random effects (*B* and *C*). Coefficients for the fixed-effect parameters are separated from those for the random-effect parameters by the vertical bar (|). The coefficients correspond to the parameter estimates in the order in which the parameter estimates are listed in the output.

Example

Suppose that factor *A* has three levels and factor *B* has four levels.

```
MIXED Y BY A B
  /FIXED = A B
  /TEST = 'test example' A 1 -1 0 DIVISOR=3;
                  B 0 0 1 -1 DIVISOR=4.
```

- For effect *A*, all contrast coefficients will be divided by 3; therefore, the actual coefficients are (1/3,–1/3,0).
- For effect *B*, all contrast coefficients will be divided by 4; therefore, the actual coefficients are (0,0,1/4,–1/4).

Interpretation of Random Effect Covariance Structures

This section is intended to provide some insight into the specification random effects and how their covariance structures differ from versions prior to 11.5. Throughout the examples, let *A* and *B* be factors with three levels, and let *X* and *Y* be covariates.

Example (Variance Component Models)

Random effect covariance matrix of *A:*

$$G_A = \sigma_A^2 I_3$$

Random effect covariance matrix of *B:*

$$G_B = \sigma_B^2 I_3$$

Overall random effect covariance matrix:

$$G = \begin{bmatrix} G_A & 0 \\ 0 & G_B \end{bmatrix}_{6\times 6}$$

Prior to version 11.5, this model could be specified by:

```
/RANDOM = A B | COVTYPE(ID)
```

or

```
/RANDOM = A | COVTYPE(ID)
/RANDOM = B | COVTYPE(ID)
```

with or without the explicit specification of the covariance structure.

As of version 11.5, this model could be specified by:

```
/RANDOM = A B | COVTYPE(VC)
```

or

```
/RANDOM = A | COVTYPE(VC)
/RANDOM = B | COVTYPE(VC)
```

with or without the explicit specification of the covariance structure.

or

```
/RANDOM = A | COVTYPE(ID)
/RANDOM = B | COVTYPE(ID)
```

with the explicit specification of the covariance structure.

Example (Independent Random Effects with Heterogeneous Variances)

Random effect covariance matrix of *A:*

$$G_A = \begin{bmatrix} \sigma_{A1}^2 & 0 & 0 \\ 0 & \sigma_{A2}^2 & 0 \\ 0 & 0 & \sigma_{A3}^2 \end{bmatrix}$$

Random effect covariance matrix of B:

$$G_B = \begin{bmatrix} \sigma^2_{B1} & 0 & 0 \\ 0 & \sigma^2_{B2} & 0 \\ 0 & 0 & \sigma^2_{B3} \end{bmatrix}$$

Overall random effect covariance matrix:

$$G = \begin{bmatrix} G_A & 0 \\ 0 & G_B \end{bmatrix}_{6\times 6}$$

Prior to version 11.5, this model could be specified by:

```
/RANDOM = A B | COVTYPE(VC)
```

or

```
/RANDOM = A | COVTYPE(VC)
/RANDOM = B | COVTYPE(VC)
```

As of version 11.5, this model could be specified by:

```
/RANDOM = A B | COVTYPE(DIAG)
```

or

```
/RANDOM = A | COVTYPE(DIAG)
/RANDOM = B | COVTYPE(DIAG)
```

Example (Correlated Random Effects)

Overall random effect covariance matrix; one column belongs to X and one column belongs to Y.

$$G = \begin{bmatrix} \sigma^2 & \sigma^2\rho \\ \sigma^2\rho & \sigma^2 \end{bmatrix}$$

Prior to version 11.5, it was impossible to specify this model.

As of version 11.5, this model could be specified by:

```
/RANDOM = A B | COVTYPE(CSR)
```